
Abstract 
There are times when you may want to select a smaller subset of data to work with.  For instance, you have a very 
large data set, and you want to perform some analysis on it.  However, you may be constrained by resources.  Or, 
you may want to select a subset for auditing.  Both of these situation requires that you obtain a subset of you original 
data that is a good representation of that original data – a random sample.  This paper presents techniques for 
creating random samples.  Starting with a baseline method, successive  improvements will be made until a final highly 
efficient method is considered. 
 
Introduction 
Pulling a random sample from a data set can be helpful in the following situations: 
 

• Dealing with large data sets 
• Resource constraints 
• Data auditing 

 
Working with large data sets can be difficult.  In some cases, it may be beneficial to pull a sample of the data and 
work with it instead.  In other situations, you may want to pull a sample of records for a data auditing exercise.  In both 
cases, you will need a sample of the original data that is representative of the original.  There are situations where 
creating a random sample is overkill.  In these cases, you might be best served by just taking, say, the first 100 
records.  Or you could select every 10th record, for example.  But remember, that these ‘samples’ may not be 
representative of the entire file.  However there is increased programmer cost associated with implementing more 
robust coding solutions.  If you are pulling a one time sample and have no reason to believe that you will ever pull 
another sample from that particular (or similar) file, then implementing a simple routine may be just fine and the 
programmer time required to implement the ‘best’ algorithm may not be offset by increased program efficiency.  On 
the other hand, if you are developing a system that will run many times (perhaps even in a production type 
environment), then it is usually to your benefit to take the time and implement the ‘best’ algorithm possible.  Keeping 
these thoughts in mind, this paper assumes that we are seeking to implement the ‘best’ algorithm (in terms of 
resource utilization) for randomly selecting observations. 
 
There are several techniques available for pulling random samples.  This paper does not discuss every possibility, but 
instead starts with some basic algorithms and demonstrates how we can improve upon them.  One important 
distinction must be made up front.  The algorithms that will be discussed are grouped into two basic types – 
sequential and non-sequential.  Sequential algorithms are those that step through your data sets one record at a time 
starting with the first and processing all the way to the last.  These algorithms are useful for data sets stored on either 
tape or disk.  Non-sequential algorithms use random access techniques to select observations regardless of order.  
Random access is allowable only for data sets stored on disk.  Also, when discussing samples, there is a distinction 
between taking a sample with or without replacement.  Samples that are drawn with replacement allow for a particular 
record to be chosen more than once.  Under without replacement, each record has only one opportunity to be 
chosen.  Sequential methods are by default without replacement methods.  This makes sense.  Because of their 
‘sequentialness’, the decision to include a record into a sample must be made at the time the record is processed.  
You cannot go back and reconsider a record that has already been passed.  Non-sequential methods are by default 
with replacement methods.  A record can be considered for inclusion in the sample more than once – it is never 
passed up. 
 
We’ll start first with sequential algorithms. 
 
Sequential algorithms 
Sequential algorithms are those that step through a data step one record at a time, starting with the first record and 
stopping when the last record has been processed (unless the data step is stopped prematurely).  The point is, that 
each record is processed sequentially – in the order that they are stored in the data set. 

The following sampling methods will be discussed: 
 

1. Baseline 
2. Arrays 
3. Sorted Arrays 

 
 *------------------------------------; 
 * 1. sequential baseline             ; 
 *------------------------------------; 
 data random; 
   set largedata; 
   if ranuni(1022) > .9999 then output; 
 run; 
 proc print data=random; 
   title1 "baseline sequential method"; 
 run; 
 
As mentioned, we must spin thru the entire data set making a decision on each record to obtain our sample.  Also, we 
must decide ahead of time what value to use as comparison against the RANUNI() function.  This value could be 
parmed in, or, if our data is stored on disk, it could be calculated based on the size of the data set as follows: 
 
 *-------------------------------------------; 
 * estimating the ranuni comparison value    ; 
 * disk only – creating a sample of approx.  ; 
 * 100,000 records                           ; 
 *-------------------------------------------; 
 data _null_; 
   if 0 then set largedata nobs=total; 
   call symput(‘total’,total); 
 run; 
 %let numrecs=100000; 

 
Now, we can calculate the RANUNI comparison value as follows: 
 
  ((&total-&numrecs)/&total) 
 
One issue with this method is that we probably will not get exactly 100,000 records in the final random data set.  
However, this is just a minor inconvenience.  The real problem with this method is that we must spin thru the entire 
data set, calculating a RANUNI() value for each record to determine if that record will be included.   
 
Is there a better way?  Do we have to calculate the RANUNI() function for every record?  Let’s look at another method 
that uses arrays to store the random record values that we want. 
 
 *------------------------------------; 
 * 2. sequential arrays               ; 
 *------------------------------------; 
 
 *------------------------------------; 
 * grab # of obs – disk only          ; 
 * you can also parm this in via %let ; 
 *------------------------------------; 
 data _null_; 
   if 0 then set largedata nobs=total; 
   call symput('total',total); 
 run; 
 %*let total=1000000; 
 
 *-----------------------; 
 * set number of records ; 
 * desired in sample     ; 
 *-----------------------; 
 %let numrecs=100000; 
 
 *-----------------------; 
 * spin data and select  ; 
 *-----------------------; 
 data random; 
   set largedata; 
 
   *--------------------------; 
   * drop unwanted vars       ; 
   *--------------------------; 
   drop i count; 

   *--------------------------; 
   * create array to hold     ; 
   * record numbers and fill  ; 
   * on first pass            ; 
   *--------------------------; 
   array nums{&numrecs} _temporary_; 
   if _n_=1 then do; 
     do i=1 to &numrecs; 
       nums{i}=round(ranuni(1022)*&total,1); 
     end; 
   end; 
 
   *--------------------------; 
   * find record # and output ; 
   *--------------------------; 
   do i=1 to &numrecs; 
     if _n_=nums{i} then do; 
       output; 
       count+1; 
       leave; 
     end; 
   end; 
 
   *--------------------------; 
   * stop after we have all   ; 
   * elements in array        ; 
   *--------------------------; 
   if count=&numrecs then stop; 
 run; 
 proc print data=random (obs=100); 
   title1 "sequential arrays method"; 
 run; 
 
The first thing that you’ll notice is that we need to get the number of observations in the data set from which we are 
pulling the sample.  If the data set resides on disk then we can grab that information from the data set metadata as 
shown in the code above.  Otherwise we will need to know (or at least have a good approximation) of the size of the 
data set and parm that number in directly.  What is the improvement with this method?  Well, now we are only 
calculating the RANUNI() function equal to the number of records that we want – not for every record!  On the first 
pass through the data (_N_ = 1 loop) we are filling an array with random numbers that we will later use to select 
records.  Also, another advantage to this method is, unlike the baseline method, we can easily modify the code to 
allow for with replacement sample selection.  All we would need to do is comment out the LEAVE statement in the 
final loop. 
  
One major drawback to this method is that we must loop over the array for every record looking for a match – this can 
be a very time consuming operation!    Another potential drawback involves the use of an array.  We are limited by our 
system resources as to the size of the array.  In most cases this is not an issue.  I have used the array method with 
arrays containing as many as 1 million elements. 
 
Can we improve this method?  The generated array of random records numbers are not sorted.  Wouldn’t it be nice if 
they were?  We could then take advantage of this fact during our spin thru the data.  We could simply skip unneeded 
records.  But first, we must decide upon a sorting method. 
 
Choosing a sorting method 
We will consider three sorting algorithms.  Each one has its advantages and disadvantages.  The first method is 
called the Bubble sort.  It is one of the simplest sorting routines to code and it works well enough for smaller array 
sizes (up to 1,000 elements).  However, its performance degrades rapidly when dealing with larger arrays.  It works by 
successively looping over an array, comparing adjacent elements and swapping if necessary.  On each loop, the 
largest remaining array element is ‘bubbled’ up the array to its proper spot. 
 
Another sorting routine is the Insertion sort.  Performance wise, it is about twice as fast as the bubble sort.  It works 
by moving successively over an array, starting with the second array element and ‘inserting’ that element into its 
proper spot amongst the elements already considered. 
 
The final sorting method is the non-recursive QuickSort.  It is easily the most efficient sorting method of the three we 
are considering, but it also is the most difficult to code.  It works by dividing the array to sort into two subarrays where 
all elements in one subarray are less than some pivot value and the other subarray elements are greater than or 
equal to the pivot value.  This continues until a subarray of one is reached which is by definition sorted. 
 
The table below shows the three sorting methods and their runtimes: 

Comparison of sorting routines:   
      
Sample Size Bubble Insertion QuickSort

100 0.01 0.01 0.01 
1,000 0.40 0.17 0.02 
5,000 9.77 4.08 0.07 

10,000 39.00 16.23 0.10 
25,000 245.39 102.61 0.25 
50,000 977.54 409.96 0.54 

      
CPU and Wall times are in seconds 
Using the SAS system for PCs 
8.2      
 
From the table above, you see that the QuickSort algorithm really does perform well for even very large arrays.  When 
choosing a sorting routine, unless you have a good reason to implement another, I would suggest implementing the 
QuickSort.  The sorted array code below uses QuickSort but includes commented out code for both the Bubble and 
Insertion sorts.  Lets take a look: 
 
 *------------------------------------; 
 * 3. sequential sorted arrays        ; 
 *------------------------------------; 
 
 *------------------------------------; 
 * grab # of obs – disk only          ; 
 * You can also parm this in via %let ; 
 *------------------------------------; 
 data _null_; 
   if 0 then set largedata nobs=total; 
   call symput('total',total); 
 run; 
 %*let total=1000000; 
 
 *-----------------------; 
 * set number of records ; 
 * desired in sample     ; 
 *-----------------------; 
 %let numrecs=100000; 
 
 data random; 
   set largedata; 
 
   *--------------------------; 
   * drop unwanted vars       ; 
   *--------------------------; 
   drop i spot; 
 
   *--------------------------; 
   * create array to hold     ; 
   * record numbers and fill  ; 
   * on first pass            ; 
   *--------------------------; 
   if _n_=1 then do; 
     array nums{&numrecs} _temporary_; 
     do i=1 to &numrecs; 
       nums{i}=round(ranuni(1022)*&total,1); 
     end; 
 
     *--------------------------; 
     * sort array using bubble  ; 
     * sort                     ; 
     *--------------------------; 
     /* 
     drop temp j;   
     do i=1 to &numrecs-1; 
       do j=1 to &numrecs-i; 
         if nums{j} > nums{j+1} then do; 
           temp=nums{j}; 
           nums{j}=nums{j+1}; 

           nums{j+1}=temp; 
         end; 
       end;  
     end; 
    */ 
 
     *----------------------------; 
     * sort array using insertion ; 
     * sort                       ; 
     *----------------------------; 
     /* 
     drop temp j; 
     do i=2 to &numrecs; 
       temp = nums{i}; 
       j=i; 
       do while (nums{j-1} > temp); 
         nums{j} = nums{j-1}; 
         j = j-1; 
         if j <=1 then leave; 
       end; 
       nums{j} = temp; 
     end; 
     */ 
 
     *--------------------------; 
     * sort array using quick   ; 
     * sort                     ; 
     *--------------------------; 
     drop stackPointer first last left right true pivot temp; 
     array stack{64} _temporary_; 
     stack{1}=1; 
     stack{2}=&numrecs; 
     stackpointer=3; 
     do while (stackpointer ne 1); 
       stackpointer=stackpointer-2; 
       first=stack{stackpointer}; 
       last=stack{stackpointer+1}; 
       do while (first < last); 
         pivot = nums{int((first+last)/2)}; 
         left = first-1; 
         right = last+1; 
         true=1; 
         do while (true); 
           do until (nums{right} le pivot); 
             right=right-1; 
           end; 
           do until (nums{left} ge pivot); 
             left=left+1; 
           end;             
           if left ge right then leave; 
           temp = nums{left}; 
           nums{left} = nums{right}; 
           nums{right} = temp; 
         end; 
         if (right-first) lt (last-right) then do; 
           stack{stackpointer} = right+1; 
           stack{stackpointer+1} = last; 
           stackpointer=stackpointer+2; 
           last=right; 
         end; else do; 
           stack{stackpointer} = first; 
           stack{stackpointer+1} = right; 
           stackpointer=stackpointer+2; 
           first = right+1;             
         end; 
       end; 
     end; 
   end; 
 
   *--------------------------; 
   * now use information about; 
   * sorted array to speed    ; 
   * processing               ; 
   *--------------------------; 
   retain spot 1; 
   if _n_ < nums{spot} then delete; 

   else do; 
     output; 
     if spot=&numrecs then stop; 
     spot+1; 
     do while (nums{spot} = nums{spot-1}); 
       if spot=&numrecs then stop; 
       spot+1; 
     end; 
   end; 
 run; 
 proc print data=random (obs=100); 
   title1 "sequential sorted arrays method"; 
 run;  
 
We can use the sorted array along with the automatic datastep variable _N_ to simply bypass any record that is not in 
the array.  If the current record is in the array, then we output that record and increment our counter into the array - 
bypassing any duplicate record numbers.  Again, we could modify this code to allow for with replacement.  We have 
eliminated the need to loop over the array for each record searching for a match as we did in the unsorted array 
method.  The real bottom line question is how much has all this code we’ve written improved efficiency over the 
original baseline method?  Well the answer is it depends.  The larger the sample size relative to the data set size we 
are pulling the sample from, the more attractive the baseline method becomes because the effect of calculating a 
RANUNI() value for each record is lessened.  In the worst case scenario, if we were pulling a sample that we equal to 
the data set size (something we would never do), we would have to perform a RANUNI() calculation for every record 
anyway, and the time spent to sort an array would just be extra.  If on the other hand we are pulling a relatively small 
sample from a large data set, then the sorted array technique performs quite well when compared to the baseline 
method. 
 
Now, lets move on to examine the non-sequential algorithms. 
 
Non-Sequential algorithms 
Non-sequential algorithms use random access techniques to select observations regardless of order.  Using the SET 
statement’s POINT= option, you can select the 1st record, then the 100th, then the 2nd, and so on.  Random access is 
allowable only for data sets stored on disk.  What is interesting to note regarding non-sequential methods (completely 
the opposite of sequential) is that they are by nature with replacement methods and additional work needs to be done 
in order to make them without replacement.  This makes sense because once a record has been selected, it is still 
available for further consideration – we have not ‘passed’ it up as in the sequential methods.  In order to make it 
unavailable, you would need to somehow tag it (or otherwise not allow it) to be selected again.  At the end of this 
section, I’ll discuss one alteration to the final method that can be used to implement without replacement. 
 
As with the sequential methods, I will start with a baseline method then using the improvements already developed for 
the sequential algorithms move directly to a final sorted array method. 
 
The following sampling methods will be discussed: 
 

1. Baseline 
2. Sorted Arrays 

 
 *------------------------------------; 
 * 1. non-sequential baseline         ; 
 *------------------------------------; 
 
 *-----------------------; 
 * set number of records ; 
 * desired in sample     ; 
 *-----------------------; 
 %let numrecs=100000; 
 
 data random; 
   drop i; 
   do i=1 to &numrecs; 
     select=round(ranuni(1022)*total,1); 
     set largedata point=select nobs=total; 
     output; 
   end; 
   stop; 
 run; 
 proc print data=random (obs=100); 
   title1 "non-sequential baseline method"; 
 run; 

One of the first things you’ll notice is that we need to include a STOP statement in the data step.  If this statement 
were omitted then the data step would go into an infinite loop.  Using random access, we no longer are accessing 
each successive record, but are moving directly to just the records we want.  Let’s say that you data set contains 1 
million records, and you want to sample just 100,000.  Using these techniques would allow you to read just the 
100,000 as opposed to the sequential methods where you would need to read all 1 million.  In general, if your data set 
is stored on disk, you will want to use random access.   
 
Finally, lets look at the array version: 
 
 *------------------------------------; 
 * 2. non-sequential sorted arrays    ; 
 *------------------------------------; 
 
 *-----------------------; 
 * set number of records ; 
 * desired in sample     ; 
 *-----------------------; 
 %let numrecs=100000; 
 
 data random; 
 
   *--------------------------; 
   * drop unwanted vars       ; 
   *--------------------------; 
   drop i; 
 
   *--------------------------; 
   * create array to hold     ; 
   * record numbers and fill  ; 
   * on first pass            ; 
   *--------------------------; 
   if _n_=1 then do; 
     array nums{&numrecs} _temporary_; 
     do i=1 to &numrecs; 
       nums{i}=round(ranuni(1022)*&total,1); 
     end; 
     *--------------------------; 
     * sort array using quick   ; 
     * sort                     ; 
     *--------------------------; 
     drop stackpointer first last left right true pivot temp; 
     array stack{64} _temporary_; 
     stack{1}=1; 
     stack{2}=&numrecs; 
     stackpointer=3; 
     do while (stackpointer ne 1); 
       stackpointer=stackpointer-2; 
       first=stack{stackpointer}; 
       last=stack{stackpointer+1}; 
       do while (first < last); 
         pivot = nums{int((first+last)/2)}; 
         left = first-1; 
         right = last+1; 
         true=1; 
         do while (true); 
           do until (nums{right} le pivot); 
             right=right-1; 
           end; 
           do until (nums{left} ge pivot); 
             left=left+1; 
           end;             
           if left ge right then leave; 
           temp = nums{left}; 
           nums{left} = nums{right}; 
           nums{right} = temp; 
         end; 
         if (right-first) lt (last-right) then do; 
           stack{stackpointer} = right+1; 
           stack{stackpointer+1} = last; 
           stackpointer=stackpointer+2; 
           last=right; 
         end; else do; 
           stack{stackpointer} = first; 
           stack{stackpointer+1} = right; 
           stackpointer=stackpointer+2; 

           first = right+1;             
         end; 
       end; 
     end; 
   end; 
 
   *--------------------------; 
   * use random access        ; 
   *--------------------------; 
   do i=1 to &numrecs; 
     spot=nums{i}; 
     set largedata point=spot; 
     output; 
   end; 
   stop; 
 run; 
 proc print data=random (obs=100); 
   title1 "non-sequential sorted array method"; 
 run; 
 
What are the improvements with this method?  Again, we create an array with the random record numbers that we 
want, then we sort that array.  But this time, taking a look at the two code segments above, it appears that this 
additional code just increases our overhead.  Looking at the final looping code that actually selects records you’ll 
notice that they are fairly similar.  The only difference is that the Baseline code creates the random record numbers 
inside the loop, while the Arrays method creates and sorts them prior to the loop then just references them.  Each 
method performs the same number of RANUNI() calculations.  So why do we go through the trouble of sorting the 
array?  To answer this, take a look at the following table that shows the run times associated with selecting varying 
array sizes with and without sorting: 
 

With and without sorting arrays comparison     
        
Data set size: 10,000,000  records     
        

Sample <== With Sort ==> <== Without Sort ==> 
Size CPU Wall CPU Wall   
1,000 0.07 11.40 0.06 11.40   

10,000 0.26 14.40 0.58 23.40   
25,000 0.38 14.40 1.44 42.00   
50,000 0.59 14.40 2.64 74.40   
100,000 1.08 15.60 5.17 163.80   
250,000 2.56 16.20 12.79 332.40   
500,000 5.04 18.60 24.60 625.20   

1,000,000 10.20 25.20 52.34 1302.00   
        
        
CPU and Wall times are in seconds     
Using Version 8.2 running under the OS/390 operating system on MVS 
 
As the table shows, we realize time savings as the size of the array increases.  This is because using a sorted array 
decreases or eliminates disk thrashing.  Disk thrashing occurs as the disk head moves to read each record on the 
data set.  In the worst case scenario, when using an unsorted array, the head might have to read the first record, then 
move all the way to the last record and read it, then move to the second and read that one, and so on, and so on.  
Using a sorted array allows the disk head to move sequentially through the records.  The amount and severity of 
thrashing depends on two things – one, the size of the data set and two, the number of array elements requested.  In 
the table above, we locked down the data set size and allowed the array size to change.  But keep in mind that 
thrashing (or at least the impact of thrashing on wall time) is reduced with smaller data sets as the disk head does not 
have to move as far.  Also note from the table that the real difference in time shows up in wall time.  The amount of 
work that the two algorithms perform is not really that disparate.  The difference in wall time is due to disk thrashing 
which is not measured as part of CPU time. 

One final thought.  The final Arrays code above is a  with replacement sampling method.  It would be fairly easy to 
modify the code to allow for non-replacement.  After sorting, code would be inserted that checks for and replaces any 
duplicate record numbers.  The array would then need to be resorted.  This would continue until all duplicates are 
eliminated.  The following pseudo-code shows these modifications: 
 
   *--------------------------; 
   * Create array with obs    ; 
   * on first observation     ; 
   * sort it                  ; 
   *--------------------------; 
   if _n_=1 then do; 
 
     *--------------------------; 
     * fill array               ; 
     *--------------------------; 
     [ see fill array code above ] 
 
     *--------------------------; 
     * start loop               ; 
     *--------------------------; 
     dupfound=1; 
     do while (dupfound); 
 
       *--------------------------; 
       * sort array               ; 
       *--------------------------; 
       [ see QuickSort code above ] 
 
 
       *--------------------------; 
       * search for dupes in array; 
       * and replace              ; 
       *--------------------------; 
       dupfound=0; 
       current = nums{1}; 
       do i=1 to &numrecs-1; 
         if current = nums{i+1} then do; 
           dupfound=1; 
           nums{i+1} = round(ranuni(1022)*&total,1); 
         end; else current=nums{i+1}; 
       end; 
     end;  
   end; 
 
Conclusion 
Creating a random sample can be helpful in many situations.  There are many ways to pull a sample.  The method 
used depends on the underlying data source of the sample – sequential vs. non-sequential, the size of the sample 
needed, and sampling needs – a one time sample vs. repeated samples over time.  Once this information is 
determined, the best method can be applied. 

